```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

We consider the following matrix of differential operators with polynomial coefficients appearing in the study of the Beltrami equation $\mathrm{div}\,((1/x_1)\,\mathrm{grad}\,u) = 0$:

```
> A:=DefineOreAlgebra(diff=[d[1],x[1]],diff=[d[2],x[2]],polynom=[x[1],x[2]]):
```

```
> R:=matrix(2,2,[d[1],-x[1]*d[2],d[2],x[1]*d[1]]);
```

$$R := \begin{bmatrix} d_1 & -x_1\,d_2 \\ d_2 & x_1\,d_1 \end{bmatrix}$$

Let us denote by A the Weyl algebra $A_2(\mathbb{Q})$ of differential operators with polynomial coefficients and $M = A^{1\times 2}/(A^{1\times 2}\,R)$ the left A-module finitely presented by R. As A is a non-commutative ring and M is not a finitely generated $\mathbb{Q}$-vector space, we know that the endomorphism ring $E = \mathrm{end}_A(M)$ of M is just an abelian group and an infinite-dimensional $\mathbb{Q}$-vector space. Hence, we can only compute the endomorphisms of M defined by means of matrices whose entries have a fixed order and a fixed degree. Let us find the matrices P of first order in the d_i's and first degree in the x_i's defining endomorphisms of M:

```
> Endo:=Morphisms(R,R,A,1,1);
```

$$Endo := [\begin{bmatrix} a_5\,x_2\,d_2 + a_5\,x_1\,d_1 + a_4\,d_2 - a_5 + a_3 & 0 \\ -a_5\,d_2 & a_5\,x_2\,d_2 + a_4\,d_2 + a_3 \end{bmatrix}, [Ore\_algebra,$$
$$[\text{“}diff\text{''}, \text{“}diff\text{''}], [x_1, x_2], [d_1, d_2], [x_1, x_2], [a_1, a_2, a_3, a_4, a_5, a_6, a_7], 0, [], [], [x_1, x_2], [], [],$$
$$[diff = [d_1, x_1], diff = [d_2, x_2]]]]$$

We obtain that an element $e \in E$ is defined by $e(\pi(\lambda)) = \pi(\lambda\,Endo[1])$, for all $\lambda \in A^{1\times 2}$, where $\pi : A^{1\times 2} \longrightarrow M$ denote the projection onto M.

Let us now search for idempotents of E defined by matrices of the form $Endo[1]$:

```
> Idem:=IdempotentsMat(R,Endo[1],A,Endo[2]);
```

$$Idem := [\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}]$$

We only find the two trivial idempotents of E, namely, 0 and id_M. However, we can try to find an element of E which is homotopically equivalent to 0 which allows us to prove that the matrix R is equivalent to a matrix of the form $\mathrm{diag}(1, a)$, where a is a certain element of A:

```
> P:=Idem[1]; Q:=evalm(P); Z:=diag(0$2);
```

$$P := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix} \quad Q := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix} \quad Z := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

Let us search for solutions of the algebraic Riccati equation $\Lambda\,R\,\Lambda = \Lambda$, where Λ is a first degree polynomial matrix in the x_i's:

```
> Mu:=Riccati(R,P,Q,Z,A,0,1,alpha);
```

$$Mu := [[\begin{bmatrix} x_1 & \alpha_1\,x_1 \\ \alpha_1 & -1 \end{bmatrix}, \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}], [Ore\_algebra, [\text{“}diff\text{''}, \text{“}diff\text{''}], [x_1, x_2], [d_1, d_2], [x_1, x_2],$$
$$[\alpha_1], 0, [], [{\alpha_1}^2 + 1], [x_1, x_2], [], [], [diff = [d_1, x_1], diff = [d_2, x_2]]]]$$

```
> Lambda:=Mu[1,1];
```

$$\Lambda := \begin{bmatrix} x_1 & \alpha_1 x_1 \\ \alpha_1 & -1 \end{bmatrix}$$

The matrix Λ admits R as a generalized inverse over the ring $B = \mathbb{Q}[\alpha_1](\alpha_1^2+1)[d_1,d_2] = \mathbb{Q}(i)[d_1,d_2]$, i.e., we have $\Lambda R \Lambda = \Lambda$ over B.

```
> B:= Mu[2]:
```

Then, the matrices $J = P + \Lambda R$ and $K = Q + R \Lambda$ defined by

```
> J:=evalm(P+Mult(Lambda,R,B)); K:=evalm(Q+Mult(R,Lambda,B));
```

$$J := \begin{bmatrix} x_1 d_1 + \alpha_1 x_1 d_2 & -x_1{}^2 d_2 + \alpha_1 x_1{}^2 d_1 \\ \alpha_1 d_1 - d_2 & -\alpha_1 x_1 d_2 - x_1 d_1 \end{bmatrix}$$

$$K := \begin{bmatrix} 1 + x_1 d_1 - \alpha_1 x_1 d_2 & \alpha_1 x_1 d_1 + \alpha_1 + x_1 d_2 \\ x_1 d_2 + \alpha_1 x_1 d_1 & \alpha_1 x_1 d_2 - x_1 d_1 \end{bmatrix}$$

are idempotent matrices, i.e., $J^2 = J$, $K^2 = K$, which define an idempotent $e \in E$ as we have:

```
> subs(alpha[1]^2=-1,evalm(Mult(J,J,B)-J));
> subs(alpha[1]^2=-1,evalm(Mult(K,K,B)-K));
> subs(alpha[1]^2=-1,evalm(Mult(R,J,B)-Mult(K,R,B));
```

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix} \quad \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix} \quad \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

Using the fact that $J^2 = J$ and $K^2 = K$, we obtain that the A-modules $\ker_B(.J)$, $\ker_B(.K)$, $\mathrm{im}_B(.J) = \ker_B(.(I_2 - J))$ and $\mathrm{im}_B(.K) = \ker_B(.(I_2 - K))$ are projective. Let us check whether or not those left B-modules are free and, if so, let us compute bases of them:

```
> U1:=SyzygyModule(J,B): U2:=SyzygyModule(evalm(1-J),B):
> V1:=SyzygyModule(K,B): V2:=SyzygyModule(evalm(1-K),B):
> U:=subs(alpha[1]^2=-1,stackmatrix(U1,U2));
> V:=subs(alpha[1]^2=-1,stackmatrix(V1,V2));
```

$$U := \begin{bmatrix} \alpha_1 & -x_1 \\ \alpha_1 d_2 + d_1 & \alpha_1 x_1 d_1 - x_1 d_2 \end{bmatrix}$$

$$V := \begin{bmatrix} x_1 d_2 + \alpha_1 x_1 d_1 & -x_1 d_1 - 1 + \alpha_1 x_1 d_2 \\ \alpha_1 & -1 \end{bmatrix}$$

We obtain that the left B-modules $\ker_B(.J)$, $\mathrm{im}_B(.J)$, $\ker_B(.K)$ and $\mathrm{im}_B(.K)$ are free of rank 1. Therefore, the matrices U and V are two unimodular matrices over B, i.e., $U, V \in \mathrm{GL}_2(B)$.

Then, the matrix R is equivalent to the block-diagonal matrix $S = V R U^{-1}$ defined by:

```
> S:=subs(alpha[1]^2=-1,alpha[1]^3=-alpha[1],simplify(Mult(V,R,
> LeftInverse(U,B),B)));
```

$$S := \begin{bmatrix} \frac{\alpha_1 x_1 d_1{}^2 + \alpha_1 x_1 d_2{}^2 - d_2}{\alpha_1} & 0 \\ 0 & -\frac{1}{\alpha_1} \end{bmatrix}$$

If we introduce the following simple elementary matrix

```
> Y:=evalm([[1,0],[0,-alpha[1]]]);
```

$$Y := \begin{bmatrix} 1 & 0 \\ 0 & -\alpha_1 \end{bmatrix}$$

we then obtain that the matrix $T = Y\,S$ has the simple form:

```
> T:=map(collect,subs(alpha[1]=I,Mult(Y,S,B)),{d[1],d[2]},distributed);
```

$$T := \begin{bmatrix} x_1\,{d_1}^2 + x_1\,{d_2}^2 + i\,d_2 & 0 \\ 0 & 1 \end{bmatrix}$$

Hence, if $\mathcal{F}$ denotes a left B-module, then we get that $\ker_{\mathcal{F}}(R.)$ is equivalent to $(x_1\,\Delta + i\,d_2)\,\zeta = 0$.